ECE4144 – Communication Hands On Assignment

The objective of this hands-on assignment is to implement a simple communication program based on USART built on the board. The processor has only one USART, which is UART1 (instead of what written in the document). Based on the datasheet, I have implemented the following code to fulfill requirements:

```
/* UCSR1A is set as default.
Normal transmission speed,
disable the multi-processor communication mode */
UCSR1A = 0x00;            // Reset the UCSR1A

UCSR1B = 0x00;            // Reset the UCSR1B
UCSR1B |= (1 << RXCIE1) | // Enable RX Complete Interrupt
        (1 << RXEN1)  |   // Enable Receiver
        (1 << TXEN1);     // Enable Transmitter

UCSR1C = 0x00;            // Reset the UCSR1C
UCSR1C |= (1 << UCSZ11) | // Set Character Size to 8-bit
        (1 << UCSZ10);

UBRR1 = 51;               // UBRR1 = (fosc / (16 * Baud Rate)) - 1
                          //       = (8MHz / (16 * 9600)) - 1 = 51.08
```

After following the requirements in question 2 and question 3, we can have the following code:

```
#include <Arduino.h>

char receivedByte = 0;

void USART_Init() {
  /* UCSR1A is set as default.
     Normal transmission speed,
     disable the multi-processor communication mode */
  UCSR1A = 0x00;            // Reset the UCSR1A

  UCSR1B = 0x00;            // Reset the UCSR1B
  UCSR1B |= (1 << RXCIE1) | // Enable RX Complete Interrupt
            (1 << RXEN1)  | // Enable Receiver
            (1 << TXEN1);   // Enable Transmitter

  UCSR1C = 0x00;            // Reset the UCSR1C
  UCSR1C |= (1 << UCSZ11) | // Set Character Size to 8-bit
            (1 << UCSZ10);

  UBRR1 = 51;               // UBRR1 = (fosc / (16 * Baud Rate)) - 1
                            //       = (8MHz / (16 * 9600)) - 1 = 51.08
}

ISR(USART1_RX_vect) {
  receivedByte = UDR1;      // Read the received byte

}

void TransmitString(const char* str, uint8_t length) {
  // Transmit byte by byte
  for (uint8_t i = 0; i < length; i++) {
    while (!(UCSR1A & (1 << UDRE1))) {
```

```
      // Wait for the transmit buffer to be empty
    }
    UDR1 = str[i]; // Transmit the byte
  }
}

char GetNextReceivedByte(){
  char byte;
  cli();               // Disable global interrupts
  byte = receivedByte; // Read the received byte
  receivedByte = 0;    // Clear the received byte after reading
  sei();               // Enable global interrupts
  return byte;
}

void setup() {
  USART_Init(); // Initialize USART1
  sei();        // Enable global interrupts
}

void loop() {
  char currentByte = GetNextReceivedByte();
  // Check if a byte is received
  if (currentByte != 0) {
    switch (currentByte) {
      case '1':
        TransmitString("One\n", 4);
        break;
      case '2':
        TransmitString("Two\n", 4);
        break;
      default:
        TransmitString("Default\n", 8);
        break;
    }
  }
}
```

After uploaded the program to the board, I connected the board to the FTDI programmer like in the figure below:



Figure 1: Connection between the board and the FTDI programmer

Then, I opened the serial monitor in the Tera Term and typed along the commands. Then, I received the outputs like in the figure below:



Figure 2: Output of the program

which is as expected.